

Graphics with Processing



2021-11 シェーディングとマッピング


<http://vilab.org>

塩澤秀和

11.1* シェーディング

シェーディング(shading)


- シェーディングモデル(p.141)
 - 光の反射・材質のモデル(前回)
 - ポリゴンの明暗の計算モデル



□ プログラマブルシェーダー

- 最近のGPUはプログラム言語で内部処理を変更できる
- 水面や人の肌など対象ごとに、シェーディングをプログラム
- 最終回で少し触れる予定


シェーディングは2D化後



11.2* フラットシェーディング

フラットシェーディング(p.155)

- 各ポリゴンを単一色で描画
 - ポリゴンの代表点(例:重心)を決め,法線ベクトルを求める
 - 代表点における入射光と法線ベクトルから反射光(色)を計算
 - 面全体をその色で塗り潰す
 - 均一な平行光が平面に当たる場合は,光学的に正しい
 - 高速だがリアリティには欠ける




法線ベクトル(p.67)

平面の方程式から求める方法

$$ax + by + cz + d = 0$$

$$\vec{N} = (a, b, c)$$



曲面の場合は?

$$\left(\frac{\partial F}{\partial x}, \frac{\partial F}{\partial y}, \frac{\partial F}{\partial z} \right)$$

ポリゴンの辺(ベクトル)から求める方法

$$\vec{N} = \vec{v}_1 \times \vec{v}_2$$

単位法線ベクトル

$$\vec{n} = \vec{N} / |\vec{N}|$$

(大きさを1にする)


外積(クロス積)

$$\begin{pmatrix} y_1 z_2 - z_1 y_2 \\ z_1 x_2 - x_1 z_2 \\ x_1 y_2 - y_1 x_2 \end{pmatrix}$$

11.3* スムースシェーディング


グロー (Gouraud) シェーディング

- 頂点間の色を補間 (p.156)
 - 各頂点に法線ベクトルを設定
 - **各頂点で**入射光と法線ベクトルから反射光(色)を計算する
 - その色を2D画面上で滑らかに補間して面全体を塗り潰す
 - 処理が高速



フォン (Phong) シェーディング

- 法線ベクトルを補間 (p.157)
 - 各頂点に法線ベクトルを設定
 - 2D画面上の各画素に対応する法線ベクトルを補間して算出
 - **画素ごとに**入射光と反射光(色)を計算し、領域を塗り潰す
 - 鏡面反射の光沢をリアルに表現




11.4 バイリニア補間

バイリニア補間(p.156)

□ 2段階の線形補間

画面描画のために
ラスタライズされたポリゴン



1段階目

$$V_Q = (1 - \alpha)V_A + \alpha V_B$$

$$V_R = (1 - \beta)V_C + \beta V_D$$

2段階目

$$V_P = (1 - \gamma)V_Q + \gamma V_R$$

□ 各シェーディングでの補間処理

■ グローシェーディング

各頂点の色のR,G,B成分を
それぞれ**バイリニア補間**



各ピクセルの補間色を合成

■ フォンシェーディング

各頂点の法線ベクトルのx,y,z
成分をそれぞれ**バイリニア補間**



各ピクセルの法線ベクトルを合成


■ テクスチャマッピング(11.10) ではuv座標を**バイリニア補間**

11.5* ポリゴン曲面

ポリゴン曲面 (p.94)

□ 曲面の近似

- 曲面をポリゴンの集合で表す
- 三角形を使うことが多い(頂点が必ず同一平面上にあるから)
- 各頂点のデータとして元の曲面の法線ベクトルを持たせる
- 各ポリゴンをスムーズシェーディングで描画すると、色が滑らかに連続して曲面として見える



ポリゴンモデルでは、頂点に法線ベクトルを持たせる


【断面図】 本来なめらかな曲面を多数のポリゴンで近似

曲面の数式がない場合は周囲のポリゴンの法線ベクトルを平均化

法線ベクトルの設定

□ normal(nx, ny, nz)


- 頂点にシェーディングのための法線ベクトルを明示的に設定
- 対応するvertexの直前で使う



11.6 ポリゴン曲面の例

実は, QUAD_STRIPで描けば
Processingが滑らかに見える
法線ベクトルを設定してくれる

長方形を貼り合わせて
円柱をモデリング



```
noStroke();
beginShape(QUADS);
int N = 18; // 円周の分割数
for (int i = 0; i < N; i++) {
  float u1 = (float) i / N;
  float a1 = 2 * PI * u1;
  float x1 = 100 * cos(a1);
  float z1 = 100 * sin(a1);
  if (mousePressed) normal(x1, 0, z1);
  vertex(x1, -100, z1);
  vertex(x1, 100, z1);

  float u2 = (float)(i + 1) / N;
  float a2 = 2 * PI * u2;
  float x2 = 100 * cos(a2);
  float z2 = 100 * sin(a2);
  if (mousePressed) normal(x2, 0, z2);
  vertex(x2, 100, z2);
  vertex(x2, -100, z2);
}
endShape();
```

11.7* テクスチャマッピング(p.162)

テクスチャマッピング

□ texture=布目・模様


- 立体表面に画像(=色分布)をシールのように貼りつける
- 例) 球に世界地図を貼りつける
- 質感を表すのに効果てきめん
- テクスチャ画像の画素のことをテクセル(texel)という

□ uv座標(テクスチャ座標)

- テクスチャ画像内の2D座標
- モデリング座標と区別するため, (u, v) (または s, t)で表す

□ uvマッピング

- ポリゴンの各頂点に画像内の点(uv座標)を対応させる処理
- 画像 $(u, v) \rightarrow$ 空間 (x, y, z)



11.8 テクスチャマッピング関数

テクスチャマッピング関数

□ texture(画像)

- 画像: PImage型(第3回参照)
- テクスチャ画像の設定
- beginShape(), endShape()の中で指定する

□ vertex(x, y, z, u, v)

- 頂点(x, y, z)を追加し,そこにテクスチャ画像内の点(u, v)を対応づける
- 2Dでの画像変形にも使える
vertex(x, y, u, v)


□ textureMode(モード)

- uv座標の値の範囲
- NORMAL: 0.0~1.0に正規化
- IMAGE: 画像のピクセル座標

画像の右下隅が
u=1.0, v=1.0

□ テクスチャマッピングの活用

- [Topics]→[Textures]
- テクスチャとポリゴンの形状が違う場合,自動的に変形される
- 画像の小領域を別々のポリゴンに貼り付けることもできる
⇒ “テクスチャアトラス(地図)”



別々のポリゴンやオブジェクト

11.9 テクスチャマッピングの使用例

```
// 準備: 画像ファイル(kouji50m.jpg)を  
// あらかじめ講義ページからダウンロードして  
// スケッチのdataフォルダに入れておく  
// (メニューで Sketch → Add File...)
```

```
PImage tex;
```

画像はグローバル
変数で定義する

```
void setup() {  
  size(300, 300, P3D);  
  tex = loadImage("kouji50m.jpg");  
}
```

画像はsetupの中で
一度だけ読み込む

```
void draw() {  
  background(0);  
  translate(width/2, height/2, 0);  
  rotateY(-radians(frameCount));
```

長方形に画像texを
テクスチャマッピング

```
noStroke();  
beginShape(QUADS);  
texture(tex);  
textureMode(NORMAL);  
vertex(-20,-50, 0, 0.0, 0.0);  
vertex( 20,-50, 0, 1.0, 0.0);  
vertex( 20, 50, 15, 1.0, 1.0);  
vertex(-20, 50, 15, 0.0, 1.0);  
endShape();
```

uv座標は
0~1モード

0以上1以下で数値
を色々変更してみよ

```
fill(#ffffff, 128);  
stroke(#555555);  
beginShape(QUADS);  
vertex(-20,-50, 0);  
vertex( 20,-50, 0);  
vertex( 20, 50, -15);  
vertex(-20, 50, -15);  
endShape();
```

半透明や透過
テクスチャは
最後に描画


```
}
```

11.10 テクスチャの描画処理

テクスチャの描画処理(p.148)

□ 段階1:画面座標→uv座標

- 画面上の各ピクセルの描画時に画像のuv座標を逆算する
- 各頂点が持つuv座標の値からバイリニア補間で求める




□ 段階2:uv座標→描画色

- 求めたuv座標の周辺テクセルから補間等で描画色を決める
- 画像の拡大・変形技術と同じ

パースペクティブ補正

□ 透視投影での補正

- 視点からの距離(z)に反比例するようにテクスチャを縮めて貼る




11.11* その他のマッピング技術

バンプマッピング(p.166)


□ 凹凸を表面にマッピング

- 表面の細かい凹凸から計算した法線ベクトルの分布を貼り付け、凹凸があるような陰影をつける
- 任意の法線分布を用意しておくものは「法線マッピング」という

法線ベクトルをマッピング



表面は平らなまま



光の反射から
見える凹凸

その他のマッピング

□ 視差マッピング

- 凹凸のある表面を違う方向から見ると、でっばりの高さによって位置がずれて見える(視差)
- 視線と高さに応じて、uv座標をずらしてマッピングする

□ 投影テクスチャマッピング(p.164)

- プロジェクタで投影するように、テクスチャを貼り付ける
- 影の表現にも応用できる


□ 環境マッピング(p.168)

- 金属などへの景色の映り込みをテクスチャで表現する
- 視点から物体の表面に反射して見えるシーンをレンダリングし、その画像をテクスチャとして貼る

11.12 演習課題

課題

- 11.6の例をもとに,円筒の表面に画像をぐるりと貼り付けるプログラムを作成しなさい
 - textureModeはNORMALとし,全てのvertexに画像内の位置を表すuv座標を追加する
 - 回転などによって全表面が確認できるようにすること
 - PNG形式などで透過色がある画像を貼るのも面白い
- 提出方法
 - 画像も含めたフォルダをZIPファイルにまとめて提出すること
 - [ツール]→[スケッチをアーカイブ]



11.13 参考：オフスクリーンレンダリング

```
// いったん“隠し画面”に描いた図形を
// 画像としてポリゴンに貼り付ける例
PGraphics pg; // 隠し画面用変数

void setup() {
  size(400, 300, P3D);
  // 隠し画面を開く
  // 3つの引数の意味はsize関数と同じ
  pg = createGraphics(100, 100,
    JAVA2D);
}

void draw() {
  // 隠し画面上での描画処理
  pg.beginDraw(); // 開始
  pg.background(200, 200, 255);
  pg.translate(50, 50);
  pg.fill(240, 180, 180);
  pg.rotate(radians(frameCount));
  pg.rect(-100, -3, 200, 6);
  pg.endDraw(); // 終了
```

```
// 表示画面での処理
background(255);
lights();
translate(width / 2, height / 2, 0);
rotateY(radians(frameCount) / 4);

beginShape(QUAD_STRIP);
texture(pg); // 隠し画面を画像として使う
textureMode(NORMAL);
vertex(-100, -100, 0, 0, 0);
vertex(-100, 100, 0, 0, 1);
vertex( -50, -100, 50, 0.25, 0);
vertex( -50, 100, 50, 0.25, 1);
vertex(  0, -100, 0, 0.5, 0);
vertex(  0, 100, 0, 0.5, 1);
vertex( 50, -100, 50, 0.75, 0);
vertex( 50, 100, 50, 0.75, 1);
vertex(100, -100, 0, 1, 0);
vertex(100, 100, 0, 1, 1);
endShape();
}
```